



No. MM2520 November 1991

Harris Analog

HA-2520/22 SPICE OPERATIONAL AMPLIFIER MACRO-MODEL

Author: David Riemer

Introduction

This application note describes the SPICE macro-model for the HA-2520/22, an uncompensated high slew rate op-amp. The model was designed to be compatible with the well known SPICE program developed by the University of California in hope that most simulation software vendors follow this basic format and syntax. A schematic of the macro-model, the Spice net listings and various simulated performance curves are included. The macro-model schematic includes node numbers to help relate the SPICE listing to the schematic. The model is designed to emulate a typical rather than a worst case part. Most AC and DC parameters are simulated. DC parameters for the HA-2522 macromodel are slightly degraded from those of the HA-2520, with the AC parameters remaining equivalent. Significant poles are included to give the most accurate AC and transient simulation with minimum complexity.

Model Description

Input Stage

DP and DN represent the differential input resistance. Input bias currents are created by I1 and offset current is modeled with FA. Source VN represents the input offset voltage. C1 limits slew rate. No input parasitics due to package capacitance and lead inductance are included.

Gain Stage

G2, R2, CC, GOL, and RD simulate open loop gain. CC is the macro-model dominant pole capacitor.

Poles

The three most significant poles of the HA-2520/22 are modeled by RC networks.

Output Stage

EX1, D1 and D2 model output current limiting. IH and IL are the power supply currents. DPH, DPL and GPS vary the supply currents based on the opamps output current. DL, DH, ECC and EEE provide voltage clamping on the output to simulate the typical output voltage swing. Some effects of output parasitics due to package capacitance and inductance are lumped with the poles.

Parameters Not Modeled

To maintain a simple macro-model not all op amp parameters are modeled. Most of the parameters not modeled are listed below:

- Temperature Effects
- Differential Voltage Restrictions.
- Input Voltage and Current Noise.
- Common Mode Restrictions
- Tolerances for Monte Carlo Analysis
- Power Supply Range

Harris Corporation, Semiconductor Sector, makes no warranty, either expressed or implied, including but not limited to any warranties or merchantability and fitness for a particular purpose using this macro-model. This macro-model is provided solely and exclusively for use by Harris customers to assist them in the assessment of this product for possible applications. In no event will Harris be liable for special, collateral, incidental, or consequential damages in connection with or arising out of the use of the macro-model. Although every effort has been made to ensure accuracy of the information contained in this application note, Harris assumes no responsibility for inadvertent errors. Harris reserves the right to make changes to the product and the macro-model without prior notice.

Spice Listing

```

*
*COPYRIGHT (C) 1991 HARRIS CORPORATION
*ALL RIGHTS RESERVED
*
*HA-2520 MACROMODEL
*REV: 8-01-91
*BY: D.W. RIEMER
*
*PINOUT      +IN  -IN  VCC  VEE  OUT
*
.SUBCKT HA2520 1 2 4 5 3
.MODEL DP D IS=1E-14 N=+1.7943E+01
.MODEL DN D IS=+1.0714E-14N=+1.7943E+01
.MODEL DV D IS=+1.9894E-15N=.2
.MODEL D1 D IS=1E-9 N=1
.MODEL D2 D IS=1E-9 N=+1.0
.MODEL DX D IS=1E-20 N=+30.0
*
*INPUT STAGE
*VALUE OF SOURCE VN MODELS VIO AND
MAY BE ADJUSTED AS DESIRED.
*
VP 1 6 0
VN 2 7 +4E-03
I1 8 0 +1.9679E-07
FA 2 0 VN +3.1579E-02
DP 6 8 DP
DN 7 8 DN
FP 9 0 VP +1.6769E+04
FN 0 9 VN +1.5652E+04
GC 0 9 8 0 +1.9433E-07
GPP 9 0 4 0 +1.9433E-07
GPN 9 0 5 0 +1.9433E-07
IRX 0 9 +1.4394E-06
RT 9 0 1.0
*
*POLES
*
EP1 10 0 9 0 1.0
RP1 10 11 +3.184E+01
CP1 11 0 1E-10
EP2 12 0 11 0 1.0
RP2 12 13 +2.653E+01
CP2 13 0 1E-10
EP3 14 0 13 0 1.0
RP3 14 15 +1.99E+01
CP3 15 0 1E-10
*
*OUTPUT STAGE
*
G2 0 16 15 0 1.0
R2 16 0 +6.5577E+02
CC 16 17 +2.2E-11
GOL 17 0 16 0 +7.3545E+01
RD 17 0 +90.0
DH 17 18 DV
DL 19 17 DV
ECC 18 0 POLY 1 4 0 -2.7 1.0
EEE 19 0 POLY 1 5 0 +2.9 1.0
IH 4 0 +4.5E-03
IL 0 5 +4.5E-03
GPS 20 0 17 3 +0.9999E-01
DPH 4 20 DX
DPL 20 5 DX
D1 17 21 D1
D2 21 17 D2
EX1 21 0 POLY 2 17 0 3 0 0.0 -8.9345E-01 +1.8897
RO 17 3 +10.0
.ENDS HA2520

```


Spice Listing

```


*
*COPYRIGHT (C) 1991 HARRIS CORPORATION
*ALL RIGHTS RESERVED
*
*HA-2522 MACROMODEL
*REV: 8-01-91
*BY: D.W. RIEMER
*
*PINOUT      +IN  -IN  VCC  VEE  OUT
*
.SUBCKT HA2522 1 2 4 5 3
.MODEL DP D IS=1E-14      N=+1.4220E+01
.MODEL DN D IS=+1.0909E-14N=+1.4220E+01
.MODEL DV D IS=+1.1671E-15N=.2
.MODEL D1 D IS=1E-9       N=1
.MODEL D2 D IS=1E-9       N=+1.0
.MODEL DX D IS=1E-20      N=+30.0
*
*INPUT STAGE
*VALUE OF SOURCE VN MODELS VIO AND
MAY BE ADJUSTED AS DESIRED.
*
VP 1 6 0
VN 2 7 +5E-03
I1 8 0 +2.4045E-07
FA 2 0 VN +7.6087E-02
DP 6 8 DP
DN 7 8 DN
FP 9 0 VP +1.0979E+04
FN 0 9 VN +1.0064E+04
GC 0 9 8 0 +3.4558E-07
GPP 9 0 4 0 +3.4558E-07
GPN 9 0 5 0 +3.4558E-07
IRX 0 9 +2.0527E-06
RT 9 0 1.0
*
*POLES
*
EP1 10 0 9 0 1.0
RP1 10 11 +3.184E+01
CP1 11 0 1E-10
EP2 12 0 11 0 1.0
RP2 12 13 +2.653E+01
CP2 13 0 1E-10
EP3 14 0 13 0 1.0
RP3 14 15 +1.99E+01
CP3 15 0 1E-10
*
*OUTPUT STAGE
*
G2 0 16 15 0 1.0
R2 16 0 +6.5577E+02
CC 16 17 +2.2E-11
GOL 17 0 16 0 +5.3933E+01
RD 17 0 +90.0
DH 17 18 DV
DL 19 17 DV
ECC 18 0 POLY 1 4 0 -2.7 1.0
EEE 19 0 POLY 1 5 0 +2.9 1.0
IH 4 0 +5.0E-03
IL 0 5 +5.0E-03
GPS 20 0 17 3 +0.9999E-01
DPH 4 20 DX
DPL 20 5 DX
D1 17 21 D1
D2 21 17 D2
EX1 21 0 POLY 2 17 0 3 0 0.0 -8.5430E-01 +1.8506
RO 17 3 +10.0
.ENDS HA2522

```

Macro-Model Schematic




Model Performance



Model Performance (Continued)

SMALL SIGNAL RESPONSE



LARGE SIGNAL RESPONSE

